# Pole assignment state feedback design example using Mathematica

Nasser M. Abbasi
Nov 29, 2014

Detailed step by step showing how to design a gain vector K for single input system in state space such that the system has desired pole locations.

Define the A and B matrices for state space x'=Ax+Bu

```
a = {{0, 0, 1, 0}, {0, 0, 0, 1}, {-2, -1, 0, 0}, {1, -1, 0, 0}};
MatrixForm[a]
```

$$\begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ -2 & -1 & 0 & 0 \\ 1 & -1 & 0 & 0 \end{pmatrix}$$

```
b = {{0}, {0}, {1}, {0}};
MatrixForm[b]
```

$$\begin{pmatrix} 0 \\ 0 \\ 1 \\ 0 \end{pmatrix}$$

## convert to controllable form (we call this the target system)

```
sysOriginal = StateSpaceModel[{a, b}];
tf = TransferFunctionModel[sysOriginal];
sys = StateSpaceModel[tf, StateSpaceRealization → "Controllable"];
{A0, B0} = {Normal[sys][[1]], Normal[sys][[2]]};
MatrixForm[A0]
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ -3 & 0 & -3 & 0 \end{pmatrix}$$

```
MatrixForm[B0]
```

$$\begin{pmatrix} 0 \\ 0 \\ 0 \\ 1 \end{pmatrix}$$

## Obtain controllability matrix for original system

```
cOriginal = ControllabilityMatrix[sysOriginal];
MatrixForm[cOriginal]
```

$$\begin{pmatrix} 0 & 1 & 0 & -2 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & -2 & 0 \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

## Obtain controllability matrix for target system

```
sysTarget = StateSpaceModel[{A0, B0}];
cTarget = ControllabilityMatrix[sysTarget];
MatrixForm[cTarget]
```

$$\begin{pmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & -3 \\ 1 & 0 & -3 & 0 \end{pmatrix}$$

Obtain T, the transformation matrix

```
(transformationT = cTarget.Inverse[cOriginal]) // MatrixForm
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & -1 & 0 & 0 \\ 0 & 0 & 1 & -1 \end{pmatrix}$$

Verify T

```
(transformationT.a.Inverse[transformationT]) // MatrixForm
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ -3 & 0 & -3 & 0 \end{pmatrix}$$

define the gains as unknowns. These are what we will determine using pole assignment

```
gains = {k0, k1, k2, k3};
```

generate the closed loop state feedback A matrix using the traget A,B system

```
aClosed = A0 + B0.{gains};
MatrixForm[aClosed]
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ -3+k0 & k1 & -3+k2 & k3 \end{pmatrix}$$

Find the CharacteristicPolynomial of the above matrix

```
pClosed = CharacteristicPolynomial[aClosed, s]
```

$$3 - k0 - k1\, s + 3\, s^2 - k2\, s^2 - k3\, s^3 + s^4$$

Extract the coefficients so we can compare them to the design polynomial below

```
pClosedCoefficnets = CoefficientList[pClosed, s]
```

```
{3 - k0, -k1, 3 - k2, -k3, 1}
```

Generate the design polynomial from the desired pole locations

```
desiredEigenValues = {-1 + I, -1 + I, -1 - I, -1 - I};
pDesign = Expand[Times @@ ((s - #) & /@ desiredEigenValues)]
```

$4 + 8 s + 8 s^2 + 4 s^3 + s^4$

Extract the coefficients of the design polynomial

```
pDesignCoefficnets = CoefficientList[pDesign, s]
```

```
{4, 8, 8, 4, 1}
```

Solve for the gains by comparing coefficients of design polynomial with the closed loop polynomial

```
gainValues = {k0, k1, k2, k3} /.
  First@Solve[pDesignCoefficnets == pClosedCoefficnets, {k0, k1, k2, k3}]
```

```
{-1, -8, -5, -4}
```

Convert the gain vector to original space using the transformation T found above

```
gainValues = gainValues.transformationT
```

```
{-5, 4, -4, -4}
```

verify the original space now has designed eigenvalues with this gain vector

```
Eigenvalues[a + b.{gainValues}]
```

```
{-1 + i, -1 + i, -1 - i, -1 - i}
```

# Using packaged function for design

The above design is implemented in my package now using the function nma`getStateGainVector. Here are examples using it

```
Get["nma.m"]; (*load the package*)
```

```
a = {{0, 0, 1, 0}, {0, 0, 0, 1}, {-2, 1, 0, 0}, {1, -1, 0, 0}};
b = {{0}, {0}, {1}, {0}};
desiredEigenValues = {-1 + I, -1 + I, -1 - I, -1 - I};
gain = nma`getStateGainVector[a, b, desiredEigenValues, False]
```

```
{-5, 2, -4, -4}
```

Another example

```
a = {{0, 0, 1, 0, 2}, {0, 0, 0, 1, 1},
   {-2, 1, 0, 0, 4}, {1, -1, 0, 0, 5}, {1, -1, 0, 0, 1}};
b = {{0}, {0}, {1}, {0}, {1}};
desiredEigenValues = {-1 + I, -1 + I, -1 - I, -1 - I, -1};
gain = getStateGainVector[a, b, desiredEigenValues, False]
```

$$\left\{-\frac{9830}{2811}, \frac{7804}{2811}, -\frac{1208}{2811}, -\frac{560}{2811}, -\frac{15658}{2811}\right\}$$

## Turn on the flag to see the steps made in the design

```
gain = getStateGainVector[a, b, desiredEigenValues, True]
```

$$A=\begin{pmatrix} 0 & 0 & 1 & 0 & 2 \\ 0 & 0 & 0 & 1 & 1 \\ -2 & 1 & 0 & 0 & 4 \\ 1 & -1 & 0 & 0 & 5 \\ 1 & -1 & 0 & 0 & 1 \end{pmatrix}$$

$$B=\begin{pmatrix} 0 \\ 0 \\ 1 \\ 0 \\ 1 \end{pmatrix}$$

$$\text{A in controllable format}=\begin{pmatrix} 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \\ -4 & -2 & 2 & -2 & 1 \end{pmatrix}$$

$$\text{B in controllable format}=\begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 1 \end{pmatrix}$$

$$\text{Controllability matrix for original system}=\begin{pmatrix} 0 & 3 & 6 & 5 & 12 \\ 0 & 1 & 6 & 10 & 18 \\ 1 & 4 & -1 & 6 & 12 \\ 0 & 5 & 7 & 15 & 10 \\ 1 & 1 & 3 & 3 & -2 \end{pmatrix}$$

$$\text{Controllability matrix for target system}=\begin{pmatrix} 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 1 & 1 & -1 \\ 0 & 1 & 1 & -1 & -1 \\ 1 & 1 & -1 & -1 & 1 \end{pmatrix}$$

$$\text{transformation T matrix }=\begin{pmatrix} \frac{149}{5622} & \frac{109}{2811} & \frac{155}{5622} & -\frac{113}{2811} & -\frac{155}{5622} \\ -\frac{691}{5622} & \frac{268}{2811} & \frac{149}{5622} & \frac{109}{2811} & -\frac{149}{5622} \\ -\frac{229}{5622} & \frac{40}{2811} & -\frac{691}{5622} & \frac{268}{2811} & \frac{691}{5622} \\ \frac{2609}{5622} & -\frac{959}{2811} & -\frac{229}{5622} & \frac{40}{2811} & \frac{229}{5622} \\ \frac{767}{5622} & -\frac{269}{2811} & \frac{2609}{5622} & -\frac{959}{2811} & \frac{3013}{5622} \end{pmatrix}$$

design Characteristic Polynomial coefficients = {4, 12, 16, 12, 5, 1}

gains in target space are = {0, -10, -18, -10, -6}

gains in original space are = $\left\{-\frac{9830}{2811}, \frac{7804}{2811}, -\frac{1208}{2811}, -\frac{560}{2811}, -\frac{15658}{2811}\right\}$

$$\left\{-\frac{9830}{2811}, \frac{7804}{2811}, -\frac{1208}{2811}, -\frac{560}{2811}, -\frac{15658}{2811}\right\}$$